

SiC 파워 디바이스

열 모델의 사용 방법

SPICE 모델에서는 열에 관한 시뮬레이션을 실시하기 위한 열 모델이라 불리는 것이 있습니다. 열 모델을 이용한 시뮬레이션은 열 설계의 초기 단계에 큰 견적을 내기 위해 실시됩니다. 이 어플리케이션 노트에서는 열 모델의 사용방법에 대해 설명하고 있습니다.

열 모델의 입수

열 모델은 로옴의 홈페이지에서 다운로드 가능합니다. 사이트 검색 창에 형명을 검색 후, 검색 결과로부터 해당 형명을 클릭합니다. 개별 제품 페이지에 이동하기 때문에 「모델 과 툴」 탭의 디자인 모델에서 열 계산 모델을 다운로드하여 주십시오.

열 모델의 개요

열 모델의 일례를 Figure 1 에 표기하였습니다.

이것은 1200V Nch SiC MOSFET SCT3040KR 로, 같은 그림이 데이터 시트에도 기재되어 있습니다.

Tj 핀은 Junction 온도, Tc 핀은 Case 온도, Ta 핀은 주위 환경 온도를 나타냅니다.




Figure 1. 열 모델의 일례, Cauer 형 RC 열 회로판
ROHM 제작 SiC MOSFET : SCT3040KR

또한, 다운로드한 열 모델의 Netlist 를 Figure 3 에 표기하였습니다.

이 Netlist 에는, SiC MOSFET 의 전기 모델과 열 모델 둘 다 기재되어 있습니다. 회로도에 표기하면 Figure 2 와 같은 소자와 핀으로 구성됩니다.

또한 이 모델에서는 SiC MOSFET 의 전기 모델로 발생한 소비전력 (자기발열)을 열모델에 전달하는 기능을 포함하고 있습니다. 이 때문에, 평소대로 전기회로의 시뮬레이션을 실행하면 Tj 핀에 Junction 온도에 상응하는 전압이 발생합니다.

다음으로 간단하게 Netlist 의 내용을 보겠습니다. SiC

MOSFET 에 발생한 소비전력을 열 모델에 전달하는 기능을 설명하기 위해, Netlist 의 주요 부분만을 발췌한 회로를 Figure 4 에 표기하였습니다. 3 개의 블록으로 나뉘어 있습니다만, 블록 A 는 SiC MOSFET 부분, 블록 B 는 Body Diode 부분, 블록 C 는 열 저항의 모델이 됩니다.

Figure 3 의 Netlist 와 비교하면, 행 13 부터 42 가 블록 A, 행 44 부터 53 이 블록 B, 행 55 부터 62 가 블록 C 에 대응하고 있습니다. 행 55 의 E20 (전압 제어 전압원)과 행 56 의 R20 (저항)은, SiC MOSFET 과 Body Diode 로 발생한 소비전력을 열 저항 모델의 Tj 핀의 전류량으로써 전달하기 위한 콤포넌트 입니다. 통상 Tj 핀에는 전류원을 사용해 소비전력에 상응하는 전류치를 공급합니다만, 전류원은 출력 임피던스가 높고 초기치의 컨버전스가 나빠질 수 있어, 전압원과 출력 저항을 조합하여 저출력 임피던스의 전류원을 표현하고 있습니다. 출력 임피던스 R20 의 1kΩ은 열 저항에 대해 충분히 큰 값입니다.




Figure 2. 열 계산 모델에 포함되어 있는 소자 핀의 구성
ROHM 제작 SiC MOSFET : SCT3040KR

다음으로 SiC MOSFET 과 Body diode 에 발생하는 소비전력을 검출하는 부분을 설명하겠습니다. 이것은 Netlist 행 55 의 E20 내에 기재되어 있는 계산식으로 행해지고 있습니다. 이 부분을 식(1)로 추출합니다.

$$P_D = (I(V1) - I(V11)) \times V(1,3) + I(V2) \times V(2,3) \quad (1)$$

```

1 * SCT3040KR_T
2 * SiC NMOSFET with driver source Self-heating Thermal model
3 * 1200V 55A 40mOhm
4 * Model Generated by ROHM
5 * All Rights Reserved
6 * Commercial Use or Resale Restricted
7 * Date: 2019/07/09
8 *****D G S DS Tj Tc Ta
9 .SUBCKT SCT3040KR_T 1 2 3 4 Tj Tc Ta
10 .PARAM T0=25 T1=-100 T2=300
11 .FUNC K1(T) {MIN(MAX(T, T1), T2)}
12 *****
13 .FUNC R1(I, T) {30.33m*I*EXP((T-T0)/166.5)+4.822u*I*ABS(I)**1.596*EXP((T-T0)/892.6)}
14 .FUNC R2(I) {I*(1.3*(1+TANH(I))/2+3.7*(1-TANH(I))/2)}
15 .FUNC V1(I, T) {139.4m*ASINH(I/0.1277)*EXP((T-T0)/-409.0)+
16 + 559.5m*ASINH(I/3.997)*EXP((T-T0)/-1244)+25.41m*I*EXP((T-T0)/498.5)}
17 .FUNC V2(V, T) {21.45p*V**13.56*EXP((T-T0)/38.52*EXP((T-T0)/-1793))}
18 .FUNC V3(V, T) {94.67*SINH(V/94.67)*EXP((T-T0)/11811)}
19 .FUNC I1(V, W, T) {V3(V, T)*(1+3.258*(TANH((V3(V, T)-66.90)/113.2)+1)*EXP((T-T0)/-174.1)/10)*
20 + W/(ABS(W)+3.258*(TANH((V3(V, T)-66.90)/113.2)+1)*EXP((T-T0)/-174.1))}
21 .FUNC C1(U, V, W) {(559.5*V+1446*(1-576.3*TANH(W/576.3)/1.345)**-0.5204)*(1+(TANH((U+8.327)/3.706)-1)/5.183)}
22 V1 1 11 0
23 E1 11 12 VALUE={R1(I(V1), K1(V(Tj)))}
24 V2 2 21 0
25 E2 21 22 VALUE={R2(I(V2))}
26 L1 3 32 7.0n
27 R2 3 32 10
28 R5 4 32 1m
29 E6 41 0 VALUE={MIN(MAX(V(22, 32), 0), 22)}
30 V6 41 42 0
31 E7 42 43 VALUE={V1(I(V6), K1(V(Tj)))}
32 G6 43 0 VALUE={MIN(MAX(V2(MAX(V(43), 0), K1(V(Tj))), -500), 500)}
33 C6 43 0 1p
34 R6 43 0 1T
35 G1 12 32 VALUE={MIN(MAX(I1(MAX(I(V6), 0), V(12, 32), K1(V(Tj))), -300), 300)}
36 C7 12 32 1p
37 R3 12 32 1T
38 V3 22 23 0
39 C1 23 12 1p
40 G2 22 12 VALUE={I(V3)*C1(V(22, 12), MIN(MAX(V(22, 12), 0), 0.525), MIN(V(22, 12), 0))}
41 C2 22 32 1.531n
42 R4 22 3 1G
43 *****
44 .FUNC R11(I, T) {87.51m*ASINH(I/0.4297)*EXP((T-T0)/304.7)+43.40m*I*EXP((T-T0)/303.9)}
45 .FUNC I11(V, T) {17.41n*(EXP(V/0.09161/EXP((T-T0)/-311.1))-1)*EXP((T-T0)/-64.07*EXP((T-T0)/761.8))-
46 + 1u*TANH(-V/0.1)*EXP((T-T0)/180.3)-8.756E-30*(EXP(-V/25)-1)*EXP((T-T0)/-200)}
47 .FUNC C11(V, W) {832.8*(V-1.091)+1065*(1-596.7*TANH(W/596.7)/2.183)**-0.5737}
48 V11 32 51 0
49 E11 51 52 VALUE={R11(I(V11), K1(V(Tj)))}
50 V12 52 53 0
51 C11 53 1 1p
52 G11 52 1 VALUE={MIN(MAX(I11(V(52, 1), K1(V(Tj))), -300), 300)+I(V12)*C11(MAX(V(52, 1), 1.091), MIN(V(52, 1), 1.091))}
53 R11 52 1 1T
54 *****
55 E20 T0 Ta VALUE={1k*MIN(MAX(IF(TIME>0, (I(V1)-I(V11))*V(1, 3)+I(V2)*V(2, 3), 0), -2MEG), 2MEG)}
56 R20 T0 Tj 1k
57 C21 Tj Ta 1.385m
58 C22 T2 Ta 10.02m
59 C23 T3 Ta 35.72m
60 R21 Tj T2 25.59m
61 R22 T2 T3 194.8m
62 R23 T3 Tc 219.6m
63 .ENDS SCT3040KR_T

```

Figure 3. 열 모델 Netlist 의 일례
ROHM 제작 SiC MOSFET : SCT3040KR




Figure 4. SiC MOSFET 에 발생한 소비전력을 열 모델에 전달하는 기능을 설명하기 위한,

netlist 의 주요한 부분을 발췌한 회로

ROHM 제작 SiC MOSFET : SCT3040KR

제 1 항에 I(V1)은 SiC MOSFET 의 드레인-소스 간에 흐르는 전류, I(V11)은 Body Diode 에 흐르는 전류입니다. 각각을 가산한 전류치 (V11 의 극성에 주의)에 드레인-소스 간 전압 V(1,3)을 올린 드레인-소스 간의 소비전력을 계산하고 있습니다. 제 2 항은, I(V2)가 게이트-소스 간에 흐르는 전류로, V(2,3)는 게이트-소스 간 전압입니다. 양자를 올린 것으로 게이트-소스 간의 손실을 계산하고 있습니다. 제 1 항과 제 2 항의 수치를 가산함으로써 전체의 소비전력을 검출하고 있습니다.

시뮬레이션 회로의 개요

열 모델을 이용한 일반적인 시뮬레이션 회로를 Figure 5 에 표기합니다. 이때의 디바이스 구성을 Figure 6 에 표기합니다. 주위환경온도 T_A 를 바이어스 전압 V_{BIAS} 로 하여 T_A 핀에 인가, 디바이스의 소비전력 P_D 를 전류 I 로써 T_j 핀에 인가하는 것으로, RC 시정 수를 가진 전압이 T_j 핀에 발생합니다. 이 전압이 Junction 온도입니다. 또한 T_c 핀에 접속되어 있는 저항은, R1 이 케이스와 Heat sink 간의 열저항 R_{thCF} , R2 가 Heat sink 와 주위환경온도간의 열저항 R_{thFA} 입니다. 또한 R_{thCF} 는 열 계면 재료 (TIM: Thermal Interface Material)의 열 저항과 접촉 열 저항을 포함합니다. 또한 C1 은 Heat sink 의 열 용량으로, R2 와 C1 로 Heat sink 를 구성하고 있습니다.




Figure 5. 시뮬레이션 회로의 일례




Figure6. 디바이스 구성

시뮬레이션 방법

사용하는 시뮬레이터에 따라 설정방법은 상이합니다만, 여기서는 SystemVision®Cloud (Mentor Graphics Corp.)를 사용하여 설명합니다.

먼저 Netlist를 참조하여 회로 심볼을 작성합니다. Netlist의 행 8 부터 9 에 핀 정보가 적혀있습니다. 이 예로는 전기 모델의 D(Drain), G (Gate), S (Source), DS (Driver Source) 핀의 순서로 핀 번호 1, 2, 3, 4가 할당되어 있고, 또한 열 모델 Tj, Tc, Ta 핀의 순서로 핀 번호 Tj, Tc, Ta가 할당되어 있어, 합계 7핀의 회로 심볼을 작성합니다. 7핀이 갖추어져 있으면 심볼 디자인은 자유입니다. 작성 예를 Figure 7에 표기하였습니다. 컴포넌트가 2개 있습니다만, 시뮬레이션 회로가 쓰기 쉽도록 분리되어 있기 때문에, 이것으로 1개의 심볼입니다.




Figure 7. 열 모델의 회로 심볼 작성

열 모델의 Netlist는 PSpice 용에 기재되어 있어, SystemVision®Cloud에서 사용하면 구문오류가 발생합니다. 이 예에서는 Netlist의 행 55에 있는 IF 문을 다음과 같이 VALIF로 수정할 필요가 있습니다.

원문

```
E20 T0 Ta VALUE={1k*MIN(MAX(IF(TIME>0,(I(V1)-I(V11))*V(1,3)+I(V2)*V(2,3),0),-2MEG),2MEG))}
```

수정 후

```
E20 T0 Ta VALUE={1k*MIN(MAX(VALIF(TIME>0,(I(V1)-I(V11))*V(1,3)+I(V2)*V(2,3),0),-2MEG),2MEG))}
```

Figure 8이 시뮬레이션 회로입니다. 여기서는 스루 홀 패키지의 TO-247-4L (Figure 9)에 Heat sink를 장착한 상태를 상정하여 시뮬레이션을 실시합니다. 회로도 좌측 절반이 전기회로, 우측 절반이 열 모델 회로로, SiC MOSFET:SCT3040KR의 Junction 온도 Tj를 결정합니다.

전기 회로에서는 SiC MOSFET의 게이트에 신호원 vsrc2로부터 구형파를 입력하여 스위칭 동작하고 있습니다. 전원은 vsrc1부터 DC 600V가 공급되어, 45Ω의 부하저항 r1을 매개로 하여 SiC MOSFET의 드레인

접속되어 있습니다. SiC MOSFET가 On하면 드레인 전류가 약 13A 흐릅니다. 이때 SiC MOSFET의 도통저항과 드레인 전류에 의해 도통손실이 발생합니다. SiC MOSFET의 게이트와 드레인의 전압파형을 Figure 10에 표기합니다.

열 모델 회로의 Ta 핀에는 주위환경온도에 대응하는 전압원 vsrc3을 접속하여 전압 (1℃=1V)을 주게됩니다. Tc 핀은 패키지 케이스이기 때문에, r4 (=케이스와 Heat sink 간의 열 저항 RthCF)와 r5 (=Heat sink와 주위환경온도 간의 열저항 RthFA)를 통해 Ta 핀에 접속합니다. r4는 케이스와 Heat sink 간의 실리콘 구리스 등을 상정하고 있습니다. r5는 Heat sink의 열 저항입니다. r4, r5는 저항치 (1℃/W=1Ω)로 사용됩니다. c1은 Heat sink의 열 용량으로 Ta 핀에 접속합니다. r5와 c1로 Heat sink를 구성하고 있습니다. c1의 열 용량치는 Heat sink의 데이터시트를 참조 또는 식(2)으로 근사값을 사용합니다. c1의 값은 용량치 (1J/K=1F)로 사용합니다.

$$\text{Heat capacity } C = c \times \rho \times V \quad [\text{J/K}] \quad (2)$$

c : Specific heat capacity [kJ/kg · K]

ρ : Density [g/cm³]

V : Volume [cm³]

예를 들어, 알루미늄 Heat sink로, 모양이 3cm×3cm×1cm인 직육면체의 경우는 다음의 계산으로 구합니다.

Specific heat capacity of aluminum c = 0.896 [kJ/kg · K]

Density of aluminum ρ = 2.71 [g/cm³]

Volume V = 3 × 3 × 1 = 9 [cm³]

$$C = 0.896 \times 2.71 \times 9 = 21.8 \text{ [J/K]} = 21.8 \text{ [F]} \quad (3)$$

Junction 온도 (=Tj 점의 전압)를 정하기 위해, SiC MOSFET으로 발생한 소비전력을 전류로 변환하여 열 모델의 T 핀에 인가할 필요가 있습니다만, 이것은 앞과 같이 열 모델 내의 계산식(1)을 통해 이루어집니다. 또한 Tj 핀과 그라운드 간에 접속되어 있는 저항 r3은, Tj 핀이 오픈으로는 시뮬레이션 실행 시 에러가 발생하기 때문에, 계산에 영향이 없는 저항치로 종단 처리하는 목적입니다.

최종으로 Transient (과도 현상)분석을 수행하여, Tj 점의 전압을 모니터 하는 것으로 Junction 온도를 알 수 있습니다. (1V=1℃)




Figure 8. 시뮬레이션 회로



Figure 9. TO-247-4L 패키지




Figure 10. Voltage waveform of the x1 gate (below) and drain (above)

다음으로 실시 예를 소개합니다. 소자의 Junction 온도가 100℃에 도달하면 회로 동작을 정지, 온도 상승을 억제하는 회로를 넣는다고 생각할 때, 회로 동작 가능한 시간을 정하는 시뮬레이션을 실행합니다.

조건은, 형태가 3cm×3cm×1cm 인 알루미늄 Heat sink 를 붙이고, r4 는 케이스와 Heat sink 간의 실리콘 구리스로 0.1W/℃의 열 저항을 상정, r5 는 Heat sink 의 열저항으로

15℃/W 를 상정하고 있습니다. c1 은 Heat sink 의 열 용량, 앞의 식(3)으로 계산한 수로 설정하고 있습니다. 주위환경온도는 50℃입니다.

Figure11 에 결과를 보여줍니다. 회로 동작으로부터 약 350 초 후에 Junction 온도가 100℃에 도달하는 것을 알 수 있습니다.




Figure 11. Heat sink 를 붙인 상태로 Junction 온도가 100℃에 도달할 때까지의 시간을 과도 해석한 결과
열 저항 15℃/W, 모양이 3cm×3cm×1cm 인 알루미늄 Heat sink 를 상정
ROHM 제작 SiC MOSFET : SCT3040KR

SystemVision®은 Mentor Graphics Corp.의 등록상표입니다.

Notes

- 1) The information contained herein is subject to change without notice.
- 2) Before you use our Products, please contact our sales representative and verify the latest specifications :
- 3) Although ROHM is continuously working to improve product reliability and quality, semiconductors can break down and malfunction due to various factors.
Therefore, in order to prevent personal injury or fire arising from failure, please take safety measures such as complying with the derating characteristics, implementing redundant and fire prevention designs, and utilizing backups and fail-safe procedures. ROHM shall have no responsibility for any damages arising out of the use of our Products beyond the rating specified by ROHM.
- 4) Examples of application circuits, circuit constants and any other information contained herein are provided only to illustrate the standard usage and operations of the Products. The peripheral conditions must be taken into account when designing circuits for mass production.
- 5) The technical information specified herein is intended only to show the typical functions of and examples of application circuits for the Products. ROHM does not grant you, explicitly or implicitly, any license to use or exercise intellectual property or other rights held by ROHM or any other parties. ROHM shall have no responsibility whatsoever for any dispute arising out of the use of such technical information.
- 6) The Products specified in this document are not designed to be radiation tolerant.
- 7) For use of our Products in applications requiring a high degree of reliability (as exemplified below), please contact and consult with a ROHM representative : transportation equipment (i.e. cars, ships, trains), primary communication equipment, traffic lights, fire/crime prevention, safety equipment, medical systems, servers, solar cells, and power transmission systems.
- 8) Do not use our Products in applications requiring extremely high reliability, such as aerospace equipment, nuclear power control systems, and submarine repeaters.
- 9) ROHM shall have no responsibility for any damages or injury arising from non-compliance with the recommended usage conditions and specifications contained herein.
- 10) ROHM has used reasonable care to ensure the accuracy of the information contained in this document. However, ROHM does not warrants that such information is error-free, and ROHM shall have no responsibility for any damages arising from any inaccuracy or misprint of such information.
- 11) Please use the Products in accordance with any applicable environmental laws and regulations, such as the RoHS Directive. For more details, including RoHS compatibility, please contact a ROHM sales office. ROHM shall have no responsibility for any damages or losses resulting non-compliance with any applicable laws or regulations.
- 12) When providing our Products and technologies contained in this document to other countries, you must abide by the procedures and provisions stipulated in all applicable export laws and regulations, including without limitation the US Export Administration Regulations and the Foreign Exchange and Foreign Trade Act.
- 13) This document, in part or in whole, may not be reprinted or reproduced without prior consent of ROHM.



Thank you for your accessing to ROHM product informations.
More detail product informations and catalogs are available, please contact us.

ROHM Customer Support System

<http://www.rohm.com/contact/>